

159.709 Computer Graphics

Lecture 04 - Lighting & Shading

Daniel Playne

`d.p.playne@massey.ac.nz`

Visual Appearance

When rendering images of a three-dimensional scene, it is important to not only have the correct geometrical shape but also the appropriate ***visual appearance***.

One of the most important steps in achieving ***visual realism*** is to model the interactions between light and the materials of the objects in the scene.

We will see how a simple (and commonly used) lighting and surface model can be implemented using OpenGL shaders to produce more convincing rendering.

Visual Phenomena

In a physical environment there are some phenomena that are useful to understand.

- Light is emitted by a natural or artificial light source.
- Light interacts with objects in the scene - absorbed, scattered or propagated in a new direction.
- Light is absorbed by a sensor (eye, film etc).


Light

Light itself is complicated and can be modelled in a number of different ways - geometric rays, electromagnetic waves or streams of photons. Regardless of this, a **light source** is something that **emits** light.

There are many different ways that light sources can be represented in computer graphics with varying degrees of complexity. In this lecture we will cover some simple approaches.


Light Sources

The simplest light source we will use is called a ***directional light*** which models an extremely distant light source (such as the sun) and considers light to be arriving at a constant direction across the entire scene.



Light Sources

Another light source that is commonly used is a ***positional light*** which models a light source (usually as a point) within or close to the scene. For these light sources the light will be arriving from different directions depending on the position within the scene.



Light Sources

In addition to the direction the light, we also need to define the amount of illumination the light source emits.

The amount of light emitted by a light source is specified by the ***irradiance*** which is the power through a unit area surface perpendicular to the direction of the light.

This is the sum of the energy from all the photons passing through the surface in one second.

Light Sources


The light may also have a particular colour, in our simple lighting model this will consist of a simple RGB value (unlike colours, these RGB values for irradiance can be arbitrarily large).

While the RGB model is sufficient for specifying a colour, it does not correctly model how different frequencies of light may interact with different materials within a scene. There are some lighting models that try to simulate this more accurately.

Illumination


To calculate the illumination on a surface, we must compute the irradiance from all light sources at a plane parallel to that surface.

The irradiance on a surface E can be calculated as the irradiance measured perpendicular to the light direction E_L times the cosine of angle θ between $\hat{\mathbf{l}}$ (the light direction) and the normal of the surface $\hat{\mathbf{n}}$.



Surface Normals

The **surface normal** (or **face normal**) \vec{n} is a vector that is perpendicular to the surface. This is usually *normalised* to give a unit vector \hat{n} .



Surface Normals


For a flat plane or primitive such as a triangle, the face normal can be calculated as follows:

$$\vec{e}_1 = \vec{v}_1 - \vec{v}_0$$

$$\vec{e}_2 = \vec{v}_2 - \vec{v}_0$$


$$\vec{n} = \vec{e}_1 \times \vec{e}_2$$

$$\hat{n} = \frac{\vec{n}}{|\vec{n}|}$$



Surface Normals

For a curved surface, the *surface normal* is perpendicular to the tangent plane at a particular point. The normals are defined at the vertex positions of the primitives approximating the curved surface and are then interpolated across the surface.



Illumination

The calculation of irradiance is clamped to non-negative values (the case where the light comes from behind the surface). This uses a cosine function that is clamped to a minimum value of 0 which is written as $\overline{\cos}(\theta)$.

$$\begin{aligned} E &= E_L \overline{\cos}(\theta) \\ &= E_L \max(\vec{n} \cdot \vec{l}, 0) \end{aligned}$$

Where the length of \vec{n} and \vec{l} are equal to 1 and we make use of the relationship $\vec{n} \cdot \vec{l} = |\vec{n}| |\vec{l}| \cos(\theta)$.

Illumination

The irradiance on a surface from multiple light sources combine additively, this makes calculating the total irradiance on a surface as simple as adding up all of the irradiance values from each of the individual light sources:

$$E = \sum_{k=0}^n E_{L_k} \overline{\cos}(\theta_k)$$

Where E_{L_k} and θ_k are the irradiance and angle for the k^{th} light in the scene.

Materials


In most computer graphics rendering, objects are represented as surfaces. The appearance of these objects can be controlled by attaching a material to each of these surfaces that determine how the irradiance on the surface is ***absorbed*** and ***scattered***.

Scattering occurs when light encounters some discontinuity (a surface in our scenes), this does not change the amount of light but simply the direction.

Absorption happens when light is converted to some other form of energy inside matter. This does not change the direction but the amount of light.

Materials


The most important optical discontinuity is the interface between the air and the surface of an object, at this interface the light is scattered in two directions - into the surface (***refraction***) and out of it (***reflection***).



RTR 3.05.06 www.realtimerendering.com / Copyright A.K. Peters Ltd. / Fair Use

Materials


In a **transparent** object, the transmitted light continues to travel through the object. For now we will consider only objects that have multiple scattering and absorption events at (or very near) the surface and some light is re-emitted back away from the surface.



RTR 3.05.07 www.realtimerendering.com / Copyright A.K. Peters Ltd. / Fair Use

Materials

It can be seen from this illustration that the light reflected at the surface has a different direction than the light was the partially absorbed/scattered by the surface.



RTR 3.05.07 www.realtimerendering.com / Copyright A.K. Peters Ltd. / Fair Use

In our lighting calculations we will separate the surface lighting equations into two terms

- ***specular*** reflected light.
- ***diffuse*** transmitted-absorbed-scattered light.

Materials

Different materials may reflect/absorb different frequencies of light in different ratios. This is represented in our lighting model by using an RGB vector or colour for the ***specular*** and ***diffuse*** terms in our equations.

Adjusting the specular and diffuse colours for our materials allow us to represent materials of different compositions - steel, plastic, paper etc.

Phong Reflection Model

The ***Phong Reflection Model*** is a commonly used lighting model that produces a plausible approximation of lighting in a 3D scene.

Each object in a scene is assigned a material with three reflectance factors which have a different factor for red-green-blue components:

- ***ambient*** (k_a)
- ***diffuse*** (k_d)
- ***specular*** (k_s)

Phong Reflection Model

All light sources in a scene have the two intensities - ***diffuse*** and ***specular*** (also with different intensities for red-green-blue).

- ***diffuse*** (I_d)
- ***specular*** (I_s)

There is a single term I_a for the ***ambient*** light in a scene.

Phong Reflection Model

The ***ambient*** term represents light scattered around the entire scene, has no particular direction and lights all surfaces evenly.

Calculating the ambient light for a particular point on a surface is as simple as multiplying the ambient light term I_a by the ambient reflectance factor of the material k_a .

$$k_a \otimes I_a$$

(where \otimes represents a piecewise multiplication of the rgb vectors)

Phong Reflection Model

The **diffuse** term represents the light from the light sources that is scattered off the surface of the object.

The diffuse light from a point on a surface can be calculated from the light's irradiance I_d , the direction towards the light \vec{l} , the normal of the surface \vec{n} and the diffuse material k_d .

$$k_d(\vec{l} \cdot \vec{n}) \otimes I_d$$

(where \vec{l} and \vec{n} are unit vectors)

Phong Reflection Model

The ***specular*** term represents the light from the light sources that is reflected off the surface of the object.

The specular light from a point on a surface can be calculated from the light's irradiance I_d , the direction towards the light \vec{l} , the normal of the surface \vec{n} , the direction towards the viewer \vec{v} , the specular material k_s and shininess coefficient α .

$$k_s (\vec{r} \cdot \vec{v})^\alpha \otimes I_s$$


(where \vec{l} , \vec{n} and \vec{v} are unit vectors and \vec{r} is the reflection of \vec{l} in the plane defined by \vec{n})

Calculating Reflected Vector

The reflected vector \vec{r} of a vector \vec{l} in a plane defined by \vec{n} can be calculated as follows (assuming $|\vec{l}| = |\vec{n}| = 1$):

$$\vec{r} = \vec{l} - 2(\vec{l} \cdot \vec{n})\vec{n}$$


As illustrated below:



Phong Reflection Model




All put together, the Phong Reflection Model calculates the light intensity I_p at a point p on the surface of an object as

$$I_p = k_a \otimes I_a + k_d(\vec{l} \cdot \vec{n}) \otimes I_d + k_s(\vec{r} \cdot \vec{v})^\alpha \otimes I_s$$



Phong Reflection Model

Some example lighting with three different materials - brass, turquoise and black plastic.



Shading

Shading is the process of using a lighting model to calculate the outgoing radiance I along a view direction \vec{v} based on the material properties and light sources.

We will look at two shading methods and how to implement them in GLSL shaders:

- Goraud Shading
- Phong Shading

Gouraud Shading Model

The ***Gouraud Shading Model*** is a method for calculating the colour for an object in a 3D scene based on a particular lighting model.

In Gouraud Shading, the lighting is calculated per-vertex (in OpenGL it is performed in the vertex shader) and then the colour is interpolated across the primitive.

This was the default shading model used in early versions of DirectX and OpenGL as it is relatively cheap to compute (only one lighting calculation per vertex).

Gouraud Shading Model - Vertex Shader

```
1 // OpenGL 3.3
2 #version 330
3
4 // Input to Vertex Shader - Position and Normal
5 in vec4 vert_Position;
6 in vec4 vert_Normal;
7
8 // Transform Matrices
9 uniform mat4 u_Model;
10 uniform mat4 u_View;
11 uniform mat4 u_Projection;
12
13 // Light Source - Directional
14 uniform vec4 u_Light_Direction = vec4(0.0f, 0.0f, -1.0f, 0.0f);
15
16 // Light Source - Properties
17 uniform vec4 Ia = vec4(0.3f, 0.3f, 0.3f, 1.0f);
18 uniform vec4 Id = vec4(1.0f, 1.0f, 1.0f, 1.0f);
19 uniform vec4 Is = vec4(1.0f, 1.0f, 1.0f, 1.0f);
20
21 // Material - Brass
22 uniform vec4 Ka = vec4(0.329412, 0.223529, 0.027451, 1.0);
23 uniform vec4 Kd = vec4(0.780392, 0.568627, 0.113725, 1.0);
24 uniform vec4 Ks = vec4(0.992157, 0.941176, 0.807843, 1.0);
25 uniform float a = 27.89743616;
26
27 // Output to Fragment Shader
28 out vec4 frag_Colour;
```

Gouraud Shading Model - Vertex Shader

```
1 void main() {
2     // ----- Calculate Vectors -----
3     // Direction to Light (normalised)
4     vec4 l = normalize(-(u_View * u_Light_Direction));
5
6     // Surface Normal (normalised)
7     vec4 n = normalize(u_View * u_Model * vert_Normal);
8
9     // Reflected Vector
10    vec4 r = reflect(-l, n);
11
12    // View Vector
13    vec4 v = normalize(-(u_View * u_Model * vert_Position));
14
15    // ----- Calculate Terms -----
16    vec4 Ta = Ka * Ia;
17    vec4 Td = Kd * max(dot(l, n), 0.0) * Id;
18    vec4 Ts = Ks * pow((max(dot(r, v), 0.0)), a) * Is;
19
20    // Vertex Colour
21    frag_Colour = Ta + Td + Ts;
22
23    // ----- Vertex Position -----
24    gl_Position = u_Projection * u_View * u_Model * vert_Position;
25 }
```


Gouraud Shading Model

While the Gouraud Shading model does a reasonably good job of computing lighting, it does suffer from some problems:

- Loss of detail in surface centres (specular highlights cannot be centred between vertices).
- Colours are linearly interpolated rather than normals.
- Specular highlights are highly dependent on view and light directions and flash with motion.

Gouraud Shading Model

For example, the following three images are of the same object with slightly different rotations.



Phong Shading Model

The ***Phong Shading Model*** is not to be confused with the ***Phong Reflection Model*** (although they were published together and are both named after their author Bui Tuong Phong).

The ***Phong Shading Model*** is a method for calculating shading the surface of an object. In this model, the lighting is calculated *per-fragment* rather than *per-vertex* as in the Gouraud shading model.

Phong Shading Model

The Phong Shading model interpolates the surface normals between vertexes and then computes the lighting model for each individual fragment using the interpolated normal.

This has the advantage of producing more realistic surface shading (especially for smoothly curved surfaces) and avoids improper interpolation of colours.

However, it does come at the cost of being significantly more expensive to compute.

Phong Shading Model - Vertex Shader

```
1 // OpenGL 3.3
2 #version 330
3
4 // Input to Vertex Shader
5 in vec4 vert_Position;
6 in vec4 vert_Normal;
7
8 // Transform Matrices
9 uniform mat4 u_Model;
10 uniform mat4 u_View;
11 uniform mat4 u_Projection;
12
13 // Light Source - Directional
14 uniform vec4 u_Light_Direction = vec4(0.0f, 0.0f, -1.0f, 0.0f);
15
16 // Output to Fragment Shader
17 out vec4 frag_Position;
18 out vec4 frag_Normal;
19 out vec4 frag_Light_Direction;
20
21 void main() {
22     // Output to Fragment Shader - Position, Normal and Light Direction
23     frag_Position      = u_View * u_Model * vert_Position;
24     frag_Normal        = u_View * u_Model * vert_Normal;
25     frag_Light_Direction = u_View      * u_Light_Direction;
26
27     // ----- Vertex Position -----
28     gl_Position = u_Projection * u_View * u_Model * vert_Position;
29 }
```

Phong Shading Model - Fragment Shader


```
1 // OpenGL 3.3
2 #version 330
3
4 // Input from Vertex Shader
5 in vec4 frag_Position;
6 in vec4 frag_Normal;
7 in vec4 frag_Light_Direction;
8
9 // Output from Fragment Shader
10 out vec4 pixel_Colour;
11
12 // Light Source
13 uniform vec4 Ia = vec4(0.3f, 0.3f, 0.3f, 1.0f);
14 uniform vec4 Id = vec4(1.0f, 1.0f, 1.0f, 1.0f);
15 uniform vec4 Is = vec4(1.0f, 1.0f, 1.0f, 1.0f);
16
17 // Material – Brass
18 uniform vec4 Ka = vec4(0.329412, 0.223529, 0.027451, 1.0);
19 uniform vec4 Kd = vec4(0.780392, 0.568627, 0.113725, 1.0);
20 uniform vec4 Ks = vec4(0.992157, 0.941176, 0.807843, 1.0);
21 uniform float a = 27.89743616;
```

Phong Shading Model - Fragment Shader

```
1 void main () {
2     // ----- Calculate Vectors -----
3     // Direction to Light (normalised)
4     vec4 l = normalize(-frag_Light_Direction);
5
6     // Surface Normal (normalised)
7     vec4 n = normalize(frag_Normal);
8
9     // Reflected Vector
10    vec4 r = reflect(-l, n);
11
12    // View Vector
13    vec4 v = normalize(-frag_Position);
14
15    // ----- Calculate Terms -----
16    vec4 Ta = Ka * Ia;
17    vec4 Td = Kd * max(dot(l, n), 0.0) * Id;
18    vec4 Ts = Ks * pow((max(dot(r, v), 0.0)), a) * Is;
19
20    // ----- Fragment Colour -----
21    pixel_Colour = Ta + Td + Ts;
22 }
```

Transforming Normal Vectors

It is important to take note of some aspects of transforming normal vectors. The normals are direction vectors so translations should not affect them and if the scaling of the model and view matrix are all uniform then the normals can simply be transformed by the **Model** and **View** transforms.



However, if there is non-uniform scaling then the normal vectors could be distorted by the transform.

Transforming Normal Vectors

If there is non-uniform scaling in the **Model** or **View** transforms then a separate **normal** transform (N) should be computed to transform the normals separately. This can be calculated from:


$$N = (M^{-1})^T$$

The inverse of a pure rotation transform is the same as its transpose. The inverse of a scale transform is the opposite while the transpose is the same. The inverse-transpose of a combined scale and rotation matrix inverts the scale but leaves the rotation unchanged.

Blinn-Phong Reflection Model

The ***Blinn-Phong Reflection Model*** is a modified version of the Phong Reflection Model that is optimised to avoid recalculation of the reflection vector for each vertex or fragment.

Instead the Blinn-Phong Reflection model uses another *halfway* vector that is between the viewer and light-source vectors.



Blinn-Phong Reflection Model

The halfway vector $\vec{\mathbf{h}}$ can be calculated from:

$$\vec{\mathbf{h}} = \frac{\vec{\mathbf{l}} + \vec{\mathbf{v}}}{|\vec{\mathbf{l}} + \vec{\mathbf{v}}|}$$

The term $\vec{\mathbf{r}} \cdot \vec{\mathbf{v}}$ in the Phong reflection model then replaced with $\vec{\mathbf{n}} \cdot \vec{\mathbf{h}}$ to give:

$$I_p = k_a \otimes I_a + k_d(\vec{\mathbf{l}} \cdot \vec{\mathbf{n}}) \otimes I_d + k_s(\vec{\mathbf{n}} \cdot \vec{\mathbf{h}})^\alpha \otimes I_s$$

Blinn-Phong Reflection Model




The angle between the halfway vector and the surface normal is likely to be smaller than the angle between the reflected vector and the viewer (unless viewed from a very steep angle).

To correct for this, a different exponent α' can be used such that $(\vec{n} \cdot \vec{h})^{\alpha'}$ is closer to $(\vec{r} \cdot \vec{v})^{\alpha}$.

For front-light objects a value of $\alpha' = 4\alpha$ gives similar results to the Phong Reflection model.

Blinn-Phong Reflection Model

Examples of the Phong reflection model (left), the Blinn-Phong reflection model with $\alpha' = \alpha$ (centre) and the Blinn-Phong reflection model with $\alpha' = 4\alpha$ (right).



GLM - OpenGL Mathematics

OpenGL Mathematics or GLM is a header-only C++ library based on the GLSL shading language specifications. We will use this library to store and manipulate vectors and matrices on the CPU as it provides implementations for many common operations and makes manipulating these types easy.

GLM - OpenGL Mathematics

Where possible GLM is consistent with the types in GLSL shaders:

Type	Description
vec2	2-component, floating-point vector
vec3	3-component, floating-point vector
vec4	4-component, floating-point vector
ivec2	2-component, integer vector
ivec3	3-component, integer vector
ivec4	4-component, integer vector
mat2	2x2 floating-point matrix
mat3	3x3 floating-point matrix
mat4	4x4 floating-point matrix

GLM - OpenGL Mathematics

GLM provides extra functionality above and beyond GLSL features with a range of extensions as organised below:

- **GLM Core** - Only the GLSL specification.
- **GTC Extensions** - Functions and Types not specified by GLSL (stable).
- **GTX Extensions** - Experimental Functions and Types not specified by GLSL (unstable).

For this course we will stick to using **GLM Core** and the stable **GTC Extensions**.

GLM - OpenGL Mathematics

Example: building a model matrix with separate rotations around the X-axis and Y-axis.

```
1 // GLM Headers
2 #include <glm/glm.hpp>
3 #include <glm/gtc/matrix_transform.hpp>
4 #include <glm/gtc/type_ptr.hpp>
5
6 ...
7
8 // -----
9 // Model Matrix
10 glm::mat4 modelMatrix = glm::mat4(1.0f);
11 float modelThetaX = 0.8f;
12 float modelThetaY = 0.0f;
13
14 // Calculate Model Matrix
15 modelMatrix = glm::rotate(modelMatrix, modelThetaX, glm::vec3(1.0f, 0.0f, 0.0f));
16 modelMatrix = glm::rotate(modelMatrix, modelThetaY, glm::vec3(0.0f, 1.0f, 0.0f));
17
18 // Get Model Matrix location
19 GLint modelLoc = glGetUniformLocation(program, "u_Model");
20
21 // Copy Rotation Matrix to Shader
22 glUniformMatrix4fv(modelLoc, 1, GL_FALSE, glm::value_ptr(modelMatrix));
23 // -----
```


GLM - OpenGL Mathematics

Example: building a view matrix using the GLM function `lookAt`.

```
1 // -----
2 // View Matrix
3 glm::mat4 viewMatrix;
4 glm::vec3 viewPosition(0.0f, 0.0f, 5.0f);
5 glm::vec3 viewUp      (0.0f, 1.0f, 0.0f);
6 glm::vec3 viewForward (0.0f, 0.0f, -1.0f);
7
8 // Normalise Vectors
9 viewUp      = glm::normalize(viewUp);
10 viewForward = glm::normalize(viewForward);
11
12 // Construct View Matrix
13 viewMatrix = glm::lookAt(viewPosition, viewPosition + viewForward, viewUp);
14
15 // Get View Matrix location
16 GLint viewLoc = glGetUniformLocation(program, "u_View");
17
18 // Copy View Matrix to Shader
19 glUniformMatrix4fv(viewLoc, 1, GL_FALSE, glm::value_ptr(viewMatrix));
20 // -----
```

GLM - OpenGL Mathematics

Example: building a perspective projection matrix.

```
1 // -----
2 // Projection Matrix
3 glm::mat4 projectionMatrix;
4
5 // Calculate Perspective Projection
6 projectionMatrix = glm::perspective(glm::radians(67.0f), 1.0f, 0.2f, 10.0f);
7
8 // Get Projection Matrix location
9 GLint projectionLoc = glGetUniformLocation(program, "u_Projection");
10
11 // Copy Projection Matrix to Shader
12 glUniformMatrix4fv(projectionLoc, 1, GL_FALSE, glm::value_ptr(projectionMatrix));
13 // -----
```

GLM - OpenGL Mathematics

Example: creating a cube (top face)

```
1 // -----
2 // Create Cube with Positions and Normals
3 void createCube(std::vector<glm::vec4> &buffer,
4               std::vector<glm::ivec3> &indexes) {
5     // ----- Top -----
6     // Left-Top-Back - 0
7     buffer.push_back(glm::vec4(-1.0f, 1.0f, -1.0f, 1.0f));
8     buffer.push_back(glm::vec4( 0.0f, 1.0f,  0.0f,  0.0f));
9
10    // Right-Top-Back - 1
11    buffer.push_back(glm::vec4( 1.0f, 1.0f, -1.0f, 1.0f));
12    buffer.push_back(glm::vec4( 0.0f, 1.0f,  0.0f,  0.0f));
13
14    // Left-Top-Front - 2
15    buffer.push_back(glm::vec4(-1.0f, 1.0f,  1.0f, 1.0f));
16    buffer.push_back(glm::vec4( 0.0f, 1.0f,  0.0f,  0.0f));
17
18    // Right-Top-Front - 3
19    buffer.push_back(glm::vec4( 1.0f, 1.0f,  1.0f, 1.0f));
20    buffer.push_back(glm::vec4( 0.0f, 1.0f,  0.0f,  0.0f));
21
22    // Top
23    indexes.push_back(glm::ivec3(0, 2, 3));
24    indexes.push_back(glm::ivec3(0, 3, 1));
25    ...
}
```